

Gnuplot によるアニメーション

v1.3 Mar.2017

【目的】

Gnuplot で計算結果の時系列のアニメーションを行う方法を習得する。

【静止画スクリプト】

動画は時刻を変えた静止画の集まりである。そこで、まず時刻 $t=0$ における静止画を Gnuplot で描画する。次に時刻を少し進めて再び静止画を描画し、さらに時刻を進めて静止画を描画する・・・という作業を gnuplot で繰り返す作業を行う。

簡単な例として、定在波(入射波+反射波の干渉)の時間変化をアニメーションで観察する場合について説明する。入射波の振幅を 1、反射係数を Γ とすると、 x 方向の定在波(干渉波)は次式(1)で与えられる。

$$y = \underbrace{e^{j(\omega t - \beta x)}}_{\text{incident}} + \Gamma \underbrace{e^{j(\omega t + \beta x)}}_{\text{reflection}} \quad \dots (1)$$

また、その絶対値(定在波の包絡線)は次式(2)となる。

$$|y| = \left| e^{j(\omega t - \beta x)} + \Gamma e^{j(\omega t + \beta x)} \right| = \left| \underbrace{e^{j\omega t}}_{\text{大きさ!}} e^{-j\beta x} (1 + \Gamma e^{2j\beta x}) \right| = |1 + \Gamma e^{2j\beta x}| \quad \dots (2)$$

まず与えられた初期条件の下で、時刻 $t=0$ における(1)式と(2)式の静止画を描画するスクリプト(下図の静止画スクリプト)を作成する。ファイルの記述には Emacs や gedit などのエディタを使用する。

(1) スクリプトの記述と保存方法

端末を開いて次のコマンドを入力する。Emacs(イーマックス)というエディタが起動する。

```
$ emacs plane_wave.gnu animation.gnu &
```

2つのファイルに、それぞれ次ページ表3と表4の内容を記述して保存する。コメントは入力しなくてよい。保存はショートカット **Ctrl + x s** キーまたは、ファイルメニューの **上書き保存** をマウスで選択する。

簡略化のためにこのスクリプトでは、反射係数 $\Gamma \rightarrow g=-0.5$ 、周波数 $\text{freq}=10$ GHz、角速度 $\omega \rightarrow w$ 、位相定数 $\beta \rightarrow b$ と置いている。スクリプト内の変数定義方法は通常の言語と同じであるが、gnuplot では **set** コマンド(グラフのオプション設定)、**plot** コマンド(グラフ描画)、**pause -1** コマンド(Enter 入力待ち)、**load** コマンド(別ファイル読み込み)などの gnuplot 特有のコマンドや **real()**、**abs()** などの gnuplot 組み込み関数を使うことができる。

作成したスクリプトのファイル名を仮に **plane_wave.gnu** とする。(ファイル名は何でも良い)

静止画スクリプト plane_wave.gnu

```
# plane_wave.gnu (行頭の # はコメント行を表す)
g=-0.5 # Γ: 反射係数(実習事項ではこれを変化させる)
pai=acos(-1) # π: 円周率
freq=10.0e9 # f : 周波数 [Hz]
cvel=3.0e8 # c: 光速 [m/s]
w=2.0*pai*freq # ω: 角速度 [rad/s]
b=w/cvel # β: 位相定数 [rad/m]
j=sqrt(-1) # 虚数の定義
set yrange [-3:3] # グラフ Y 軸の表示範囲の設定
set xrange [-0.05:0.05] # グラフ X 軸の表示範囲の設定
set samples 1000 # グラフを構成する点の数の設定
set grid # グラフ背景にグリッド線をつける
t=0.0 # 初期時刻を 0 秒に指定する
plot real(exp(j*(w*t-b*x))+g*exp(j*(w*t+b*x))) with line, \ # (1)式の定在波(入射波+反射波)を線で描く
abs(1.0+g*exp(2.0*j*b*x)) with line lw 3 # (2)式のエンベロープを太線で描く
pause -1 'Hit "Enter key" to start animation. ... ' # Enter が押されるまで待機する
load 'animation.gnu' # animation.gnu を読み込む
← # “注” 最後の行に改行(Enter)を入れておくこと。
```

このバックスラッシュは改行記号であり、日本語キーボードの ¥ と同じ(忘れずに入力する)

【ヒント】:プロット関数の使い方

Plot 関数 1, 関数 2, 関数 3, 関数 4, ...

のように記述することで、同じ 1 つのグラフ内に複数の関数を同時にプロットすることができる。

【動画スクリプト】


次に時間を進めて静止画を再描画するスクリプト(下図の動画スクリプト)を作成する。ファイル名を仮に `animation.gnu` とする。

このスクリプトでは `replot` コマンド(直前の `plot` コマンドを再実行する), `reread` コマンド(直前の `load` コマンドを再実行する)という二つの `gnuplot` コマンドを使っている。即ち、このスクリプトで `replot` が実行されると `plane_wave.gnu` の中の `plot` コマンド行に処理が飛び、`reread` が実行されることで `plane_wave.gnu` の中の `load` コマンド行に処理が飛ぶようになっている。このため、何らかの実行停止処置をしない限り、無限に時間を進めながら描画を繰り返すことができる。

動画スクリプト `animation.gnu`

```
# animation.gnu          (行頭の # はコメント行を表す)
t=t+1/freq/50          # 時間を 1 周期の 1/50 だけ進める
replot                 # 直前に plot コマンドで描画した内容を再描画する
reread                 # 直前に load コマンドで読み込んだスクリプトを再読み込みする
<| # “注” 最後の行に改行(Enter)を入れておくこと。
```

作成した 2 ファイル `plane_wave.gnu` と `animatino.gnu` があるフォルダにおいて `gnuplot` を起動し、`plane_wave.gnu` をロードする。(下図参照)



```
kusama@typhoon:~/Program_takuma/animation/plane_wave
ファイル(F) 編集(E) 表示(V) 端末(I) タブ(B) ヘルプ(H)
[kusama@typhoon plane_wave]$ gnuplot

GNU PLOT
Version 4.0 patchlevel 0
last modified Thu Apr 15 14:44:22 CEST 2004
System: Linux 2.6.23.17-88.fc7

Copyright (C) 1986 - 1993, 1998, 2004
Thomas Williams, Colin Kelley and many others

This is gnuplot version 4.0. Please refer to the documentation
for command syntax changes. The old syntax will be accepted
throughout the 4.0 series, but all save files use the new syntax.


Type 'help' to access the on-line reference manual.
The gnuplot FAQ is available from
http://www.gnuplot.info/faq/

Send comments and requests for help to
<gnuplot-info@lists.sourceforge.net>
Send bugs, suggestions and mods to
<gnuplot-bugs@lists.sourceforge.net>


Terminal type set to 'x11'
gnuplot> load 'plane_wave.gnu'
```

これを実行すると、初期時刻 $t=0$ における定在波(赤色)と包絡線(緑色)が描画される(左図)。静止画スクリプト `plane_wave.gnu` が実行されると、`pause -1` コマンドの行で処理が一旦停止するので、停止解除するために `Enter` を入力する(右下図)。

```
Terminal type set to 'x11'
gnuplot> load 'plane_wave.gnu
Hit Enter Key, please.
```



問題がなければ、包絡線に沿って定在波が移動する様子が確認できる。アニメーションを停止するには **Ctrl^C** を押す。静止画スクリプト `plane_wave.gnu` で定義した反射係数の値(複素数でも可)を変えることで、定在波の変化を視覚的にイメージすることができる。



【参考文献】

- [1] 大竹 敢, 矢吹 道郎 : ``使いこなす gnuplot 改定第 2 版 `` テクノプレス, 2004.

以上